Journal de Théorie des Nombres de Bordeaux 6 (1994), 39–80

Implementing the Round Four maximal order algorithm

by David Ford and Pascal Letard

1. Introduction

Given p, a rational prime, and f, a monic separable polynomial in Z_p[x], let ξ = x + fQ_p[x]. Define A_f = Q_p[x]/fQ_p[x] = Q_p[ξ], and let O_f be the maximal order in A_f .

For θ ∈ A_f , let χ_θ denote the characteristic polynomial of θ and Δ_θ the discriminant of χ_θ . An element θ ∈ O_f is primary if χ_θ is congruent modulo p to a power of a polynomial ν_θ , with ν_θ monic and irreducible mod p.

The Round Four algorithm constructs an integral basis for O_f .

The algorithm has three distinct branches:

0. For α ∈ O_f with Δ_α ≠ 0, the Dedekind test (see [Cohen 1993]) applied to χ_α efficiently determines a basis for D_α, the coefficient ring of the p-radical of Z_p[α], which is an order satisfying Z_p[α] ⊆ D_α ⊆ O_f , with Z_p[α] = D_α if and only if Z_p[α] = O_f .

1. If a non-primary element of O_f is found then this element can be used to construct orthogonal idempotents and decompose the algebra as the direct sum of subalgebras of lower degree. An integral basis for the algebra can then be constructed from integral bases of the subalgebras.

2. Failing cases 0 and 1, for a given primary α ∈ O_f with Δ_α ≠ 0 a sequence of elements in O_f − Z_p[α] with increasing p-adic value can be constructed. This sequence can not be extended indefinitely without leading to case 0 or case 1.

Manuscrit reçu le 20 juillet 1993, version définitive le 3 juin 1994.

In what follows we give the theoretical underpinnings of the algorithm. Descriptions in detail of major routines (maxord, decomp, nilord, . . . ) of our Maple implementation are in Appendix I. A complete Maple listing appears as Appendix II. Experimental results are given in Appendix III.

2. Denominators Let f = a_0x^n + a_1x^{n−1} + · · · + a_{n−1}x + a_n

with a_0 = 1, and t_j = Tr(ξ^j ), so that t_0 = n. We define

A = t_0 t_1 t_2 · · · t_{n−1} t_1 t_2 t_3 · · · t_n t_2 t_3 t_4 · · · t_{n+1} . . . . . . . . t_{n−1} t_n t_{n+1} · · · t_{2n−2} .

Using Newton’s relations m∑ j=0 t_{m−j} a_j = (n − m)a_m (m ≤ n) n∑ j=0 t_{m−j} a_j = 0 (m > n) it can be shown inductively for r ≥ 0 that

rem(x^rf^′, f, x) = n∑ k=1 b_{r,k}x^{n−k} where b_{r,k} = k−1∑ j=0 t_{r+k−1−j} a_j . We define

B = b_{0,1} b_{0,2} b_{0,3} · · · b_{0,n} b_{1,1} b_{1,2} b_{1,3} · · · b_{1,n} b_{2,1} b_{2,2} b_{2,3} · · · b_{2,n} . . . . . . . . b_{n−1,1} b_{n−1,2} b_{n−1,3} · · · b_{n−1,n} .

We call a matrix in Z^{n×n}_p a lower identity if it has the form 1 ∗ ∗ · · · ∗ 0 1 ∗ · · · ∗ 0 0 1 · · · ∗ . . . . . . . . 0 0 0 · · · 1 .

For the lower identity

C = a_0 a_1 a_2 · · · a_{n−1} 0 a_0 a_1 · · · a_{n−2} 0 0 a_0 · · · a_{n−3} . . . . . . . . 0 0 0 · · · a_0 = 1 a_1 a_2 · · · a_{n−1} 0 1 a_1 · · · a_{n−2} 0 0 1 · · · a_{n−3} . . . . . . . . 0 0 0 · · · 1

we have AC = B. In essence, B is the “interesting part” of the Sylvester matrix of f and f^′. Construction of the row-reduced Hermite normal form of B gives L, T, D ∈

Z^{n×n}_p with L unimodular, T a lower identity, and diagonal

D = diag(p^{δ1} , p^{δ2} , . . ., p^{δn} )

with δ_1 ≤ δ_2 ≤ · · · ≤ δ_n, such that LB = DT. Definition. For α ∈ O_f with Δ_α ≠ 0, the reduced discriminant of χ_α is p^{dα} , given by p^{dα} Z_p = (χ_αZ_p[x] + χ^′_αZ_p[x]) ∩ Z_p.

It is clear that the reduced discriminant p^{dξ} = p^{δn} can be obtained from the (n, n) entry of the p-adic Hermite normal form of the Sylvester matrix B defined above.

Now let X = (1, ξ, . . ., ξ^{n−1}) and define C_f = {z ∈ Q_p[ξ] : Tr(zZ_p[ξ]) ⊆ Z_p}.

Clearly C_f is a Z_p-module containing O_f , and, since A = ( Tr(ξ^{j−1}ξ^{i−1}) ) , the entries of XA^{−1} form a Z_p-basis for C_f . There exist a lower identity S and diagonal E = diag(p^{−ϵ1} , p^{−ϵ2} , . . ., p^{−ϵn} ) with 0 = ϵ_1 ≤ ϵ_2 ≤ · · · ≤ ϵ_n, such that the entries of XSE form a Z_p-basis for O_f . Because p^{ϵ1+···+ϵn} is the Z_p-module index of Z_p[ξ] in O_f , we have 2(ϵ_1 + · · · + ϵ_n) ≤ δ_1 + · · · + δ_n.

Since O_f ⊆ C_f there exists a non-singular M ∈ Z^{n×n}_p for which XSE = XA^{−1}M SE = A^{−1}M = CT^{−1}D^{−1}LM DT C^{−1}SE = LM. Then H = T C^{−1}S is a lower identity and DHE = LM ∈ Z^{n×n}_p . Since (DHE)_{j,j} = p^{δj −ϵj} , 1 ≤ j ≤ n, it follows that δ_j − ϵ_j ≥ 0 (1 ≤ j ≤ n). Theorem. The diagonal entries of the p-adic Smith (or Hermite) normal form of the Sylvester matrix of f and f^′ give bounds on the successive denominators of an integral basis of A_f given in Hermite normal form. Corollary. If α ∈ O_f and Δ_α ≠ 0 then p^{dα} O_f ⊆ Z_p[α]. 3. Structural stability Assume d ≥ 0, h ∈ Z_p[x], h monic, satisfying

p^dO_f ⊆ Z_p[ξ], p^dO_h ⊆ Z_p[θ], h ≡ f (mod p^{2d}Z_p[x]) where θ = x + hQ_p[x]. Let σ : A_f → A_h be the Q_p-module isomorphism determined by σ(ξ^{j−1}) = θ^{j−1} (1 ≤ j ≤ n). For all α, β ∈ Z_p[ξ] we have σ(αβ) ≡ σ(α)σ(β) (mod p^{2d}Z_p[θ]).

Now define ~O = σ(O_f ). For any α, β ∈ O_f we have p^dα, p^dβ ∈ Z_p[ξ], so that σ(α)σ(β) = σ(αβ)+p^{−2d}(σ(p^dα)σ(p^dβ)−σ((p^dα)(p^dβ))) ∈ ~O +Z_p[θ] = ~O. Thus the Z_p-module ~O is closed under multiplication, and therefore is an order in A_h. By similar reasoning, σ^{−1}(O_h) is an order in A_f , so

O_f = σ^{−1}( ~O) ⊆ σ^{−1}(O_h) ⊆ O_f .

Consequently, ~O = O_h.

Therefore, if a Z_p-basis of O_h is known in terms of [1, θ, . . ., θ^{n−1}] then the

Qp-module isomorphism σ^{−1} immediately gives a Zp-basis of Of in terms of [1, ξ, . . ., ξ^{n−1}]. To apply this result, recall that p^dO_f ⊆ Z_p[ξ] whenever either d ≥ d_ξ or p^{2d+2} - Δ_ξ.

4. Case 1: algebraic decomposition

Assume h, a_1, a_2 ∈ Z_p[x], with

a_1h + a_2h^′ ≡ p^d (mod p^{d+1}Z_p[x]) h(ξ) ≡ 0 (mod p^mZ_p[ξ]) m ≥ 2d + 1. Taking ψ = −a_2(ξ)h(ξ)/p^m we have ψ ∈ Zp[ξ] h(ξ + p^{m−d}ψ) ≡ h(ξ) + p^{m−d}ψh^′(ξ) ≡ 0 (mod p^{m+1}Z_p[ξ]).

It follows that there exists φ ∈ Z_p[ξ] such that h(φ) = 0

Zp[φ] = Zp[ξ] φ ≡ ξ (mod p^{m−d}Z_p[ξ]). Now, using the reduced discriminant p^{dξ} of f, set m_ξ = 2d_ξ + 1.

If h ≡ f (mod p^{mξ} Z_p[x]), then A_f contains a root of h, so A_h ∼= A_f (as

Qp-algebras).

In particular, if f ≡ f_1f_2 (mod p^{mξ} Z_p[x]), f_1, f_2 monic, then A_f ∼= A_{f1f2} ∼= A_{f1} ⊕ A_{f2} . In this case an integral basis for A_f can be constructed directly by mapping integral bases of A_{f1} and A_{f2} into A_f .

(Structural stability can be applied to reduce the modulus p^{mξ} by at least one p-adic digit. But as the example f = x^2 − 5, f_1 = x − 1, f_2 = x + 1, p = 2, d_ξ = 1 shows, A_f and A_{f1f2} might then fail to be isomorphic

Qp-algebras.)

Given a non-primary element of O_f , the Round Four algorithm finds orthogonal idempotents e_1, e_2 such that e_j A_f ∼= A_{fj} , j = 1, 2, giving explicit isomorphisms ψ_j : A_{fj} → e_j A_f ψ_j (ξ_j) = e_j ξ where ξ_j = x + f_j Q_p[x], so that A_{fj} = Q_p[x]/f_jQ_p[x] = Q_p[ξ_j ].

5. Computing the p-adic GCD

Suppose f, b ∈ Z[x], with f monic. Then for any m the Maple procedure

ihermite(sylvester(f, b)) mod p^m

gives, in its last non-zero row, p^rf_1, with f_1 ∈ Z[x], f_1 monic, such that p^rf_1 ∈ fZ[x] + bZ[x] + p^mZ[x]

rem(f, f_1, x) ≡ 0 (mod p^{m−r} Z[x]).

So with f_2 = quo(f, f_1, x) mod p^m we have f ≡ f_1f_2 (mod p^{m−r} Z[x]).

Suppose further that B ∈ Z_p[x], G = gcd(f, B), b ≡ B (mod p^mZ_p[x]), and let p^sZ_p = ( f G Zp[x] +^B_GZp[x]^) ∩ Zp.

If m > s, then r = s, f_1 ≡ G (mod p^{m−s}Z_p[x]);

otherwise r < s, deg(f_1) > deg(G).

6. Idempotents: a special case

Assume e ∈ Q_p[x] with 0 < deg(e) < deg(f) and f | e(1 − e).

Choose d_r so that p^{dr} e ∈ Z_p[x], and let B_1 = p^{dr} (1 − e), B_2 = p^{dr} e.

For j = 1, 2 let G_j = gcd(f, B_j ) H_j = B_j /G_j p^{sj} Z_p = ( f G_j Z^p

[x] + B G_j Z^p [x] ) ∩ Zp so that f = G_1G_2 p^{s1} Z_p = (G_2Z_p[x] + H_1Z_p[x]) ∩ Z_p p^{s2} Z_p = (G_1Z_p[x] + H_2Z_p[x]) ∩ Z_p.

Because p^{dr} = B_1 + B_2 = G_1H_1 + G_2H_2 we have s_1 ≤ d_r, s_2 ≤ d_r.

We wish to compute p-adic approximations to gcd(f, e) and gcd(f, 1 − e).

We define B = p^{dr} (1 − e), so that B ∈ Z_p[x] and gcd(f, B) = gcd(f, 1 − e).

For m = m_r + d_r and b ≡ B (mod p^mZ_p[x]) we compute f_1, f_2 as before, giving f_1 ≡ gcd(f, 1 − e) (mod p^{mr} Z_p[x]) f_2 ≡ gcd(f, e) (mod p^{mr} Z_p[x]) f ≡ f_1f_2 (mod p^{mr} Z_p[x]).

7. Case 2: search for a power basis

For θ ∈ A_f with χ_θ = x^n + c_1x^{n−1} + · · · + c_n we define

v^∗(θ) = min_{1≤k≤n} v_p(c_k) k . The p-radical of O_f is J_f = {θ ∈ O_f : v^∗(θ) > 0}. For primary θ ∈ O_f we define D_θ = deg(ν_θ ) L_θ/M_θ = v^∗(ν_θ (θ)), with L_θ , M_θ ≥ 0, gcd(L_θ, M_θ) = 1 r_θL_θ − s_θ M_θ = 1, with r_θ > 0, s_θ ≥ 0 η_θ(θ) = ν_θ (θ)^{rθ} /p^{sθ} , so that v^∗(η_θ(θ)) = 1/M_θ.

The algorithm attempts to find α ∈ O_f with Z_p[α] = O_f .

For any α ∈ O_f with Δ_α ≠ 0 we have p^{dα} O_f ⊆ Z_p[α]. It follows that the integral elements of A_f lying outside Z_p[α] have bounded p-adic value. Assuming Z_p[α] ≠ O_f , the algorithm constructs a sequence of elements β ∈ O_f − Z_p[α] with v^∗(β) strictly increasing. Since v^∗(β) < d_α for each β, it suffices to compute β modulo p^{dα} Z_p[α].

Tests A, B, C are applied to various elements θ ∈ O_f that are produced by the algorithm.

A. If θ is non-primary, we break off the search for α and revert to case 1.

B. If D_θ - D_α then we find φ ∈ θ + Z_p[α] with either φ not primary or D_φ = lcm(D_α, D_θ) > D_α. We replace α ← φ and go to step 2.

C. If M_θ - M_α, we find a, b, c ≥ 0 such that aM_α + bM_θ − cM_αM_θ = gcd(M_α, M_θ). Then φ = α + η_α(α)^bη_θ(θ)^a/p^c satisfies D_φ = D_α and M_φ = lcm(M_α, M_θ) > M_α. So we replace α ← φ and go to step 2.

The sequence of elements β is constructed as follows.

1. Let ξ = x + f(x)Z_p[x].

Set α = ξ.

2. Apply test A to α.

If Δ_α = 0, replace α by α+kpξ for some k > 0 such that Δ_{α+kpξ} ≠ 0. Apply the Dedekind test to χ_α. If Z_p[α] = O_f the search terminates. If v^∗(α) > 0, replace α by α + 1, so that v^∗(α) = 0.

If L_α > 1, replace α by α + η_α(α), so that L_α = 1 (so η_α = ν_α, with ν_α unchanged).

We have p^{dα} O_f ⊆ Z_p[α]. We determine r such that θ^{prDα} ∈ Z_p[α] whenever θ ∈ Z_p[α] + J_f . Let θ = φ + ψ, with φ ∈ Z_p[α], ψ ∈ J_f . We have v^∗(ψ) ≥ 1/n, and ψ^k ∈ Z_p[α] whenever k ≥ nd_α. Observe that v^∗ (_ps k ) ≥ s − v^∗(k!) ≥ s − (k − 1)/(p − 1). It therefore suffices to have rD_α ≥ d_α + (k − 1)/(p − 1) for all k < nd_α.

3. Set q = p^{rDα} , with r = ⌈(d_α(n + p − 1) − 2)/(D_α(p − 1))⌉.

Set π = ν_α(α), so that v^∗(π) = 1/M_α. Set β = π^{Mα} /p.

4. Apply tests A, B, C to β.

5. Set k = M_αv^∗(β), so that v^∗(π^k) = v^∗(β). (k ∈ Zbecause M_β | M_α.)

Set γ = β/π^k, and apply tests A, B, C to γ.

6. Set δ = γ^q, and apply tests A, B, C to δ.

7. If δ ∈ Z_p[α]:

We have π^kδ ∈ Z_p[α], π^kγ = β /∈ Z_p[α], and, because D_γ | D_α, γ − δ ∈ J_f . So π^k(γ − δ) /∈ Z_p[α], but v^∗(π^k(γ − δ)) ≥ v^∗(π^k) + v^∗(γ − δ) > v^∗(π^k) = v^∗(β).

Replace β by π^k(γ − δ) = β − π^kδ, and go to step 4.

8. If δ /∈ Z_p[α]:

Then Z_p[α, γ]/J_f is not a field. We apply test A to elements of

γ + Zp[α] until discovering a non-primary element of Of .

The preceding cycle of steps, increasing v^∗(β), cannot be repeated indefinitely. It must be interrupted by one of the following:

i) discovery of α ∈ O_f with Z_p[α] = O_f ; ii) discovery of a non-primary element; iii) discovery of θ ∈ O_f with D_θ - D_α; iv) discovery of θ ∈ O_f with M_θ - M_α.

Case i) terminates the algorithm. Case ii) lowers the degree of f, and so can occur at most n − 1 times. Cases iii) and iv) occur finitely often, because D_α and M_α are bounded by n.

8. Polynomial factorization

The Round Four algorithm can readily be adapted for factorization of polynomials in Q_p[x].

First, the Dedekind test must be replaced by a test for irreducibility in

Qp[x]. It suffices to test for two conditions:

1) χ_α is irreducible in Q_p[x] if χ_α is irreducible modulo p (i.e., the image of χ_α in F_p[x] is irreducible in F_p [x]);

2) χ_α is irreducible in Q_p[x] if α is an Eisenstein element, i.e., i) α is primary, and ii) Δ_α ≠ 0, and iii) v^∗(ν_α(α)) = deg(ν_α)/ deg(χ_α).

It is a well-known theorem that

f is irreducible in Q_p[x] ⇐⇒ A_f is a field ⇐⇒ A_f contains an Eisenstein element.

Second, routines that returned integral bases must be modified to return factorizations instead.

As noted above, factorizations must be computed modulo p^{2dξ +1} (at least) to ensure that the results can be lifted to Q_p[x]; any such factorization agrees with the correct p-adic factorization modulo p^{dξ +1}.

Acknowledgements.

This work was done at the invitation of, and with generous support from, the Centre de Recherche en Mathématique de Bordeaux at Université Bordeaux I. Additional support was provided by the Natural Sciences and Engineering Research Council (Canada) and Fonds pour la Formation de Chercheurs et l’Aide à la Recherche (Québec).

References

The essential mathematical background is given in [Zariski & Samuel 1958, Ch V]. The Round Two algorithm is described in [Zimmer 1972, Ch 6(b)] and [Cohen 1993, Ch 6.1]. Elements of the Round Four algorithm are evident in [Zassenhaus 1975]. The first implementation of Round Four was [Ford 1978] (also reported in [Ford 1987]), for which the ALGEB language was created. [Böffgen 1987] gives a straightforward account (in German) but with some proofs omitted. [Böffgen & Reichert 1987] discusses application of the algorithm to factorization. [Bradford 1988] provides a wellwritten survey. [Pohst & Zassenhaus 1989, Ch 4] describes the algorithm in a general setting.

1. A. Ash, R. Pinch, R. Taylor, An bA4 extension of Q attached to a non-selfdual automorphic form on GL(3), Mathematische Annalen 291 (1991), 753–766.

2. R. Böffgen, Der Algorithmus von Ford/Zassenhaus zur Berechnung von Genzheitsbasen in Polynomalgebren, Annales Universitatis Saraviensis, Series Mathematicae 1 no. 3 (1987), 60–129.

3. R. Böffgen, M. A. Reichert, Computing the Decomposition of Primes p and p-adic Absolute Values in Semi-simple Algebras over Q, Journal of Symbolic Computation 4 (1987), 3–10.

4. R. J. Bradford, On the Computation of Integral Bases and Defects of Integrity, PhD Dissertation, University of Bath, 1988.

5. B. W. Char, K. O. Geddes, G. H. Gonnet, B. L. Leong, M. B. Monagan, S. M. Watt, Maple V Language Reference Manual, Springer-Verlag, New York, 1991.

6. H. Cohen, A Course in Computational Algebraic Number Theory, Springer-Verlag, New York, 1993.

7. D. Ford, On the Computation of the Maximal Order in a Dedekind Domain, PhD Dissertation, Ohio State University, 1978.

8. D. Ford, The Construction of Maximal Orders over a Dedekind Domain, Journal of Symbolic Computation 4 (1987), 69–75.

9. M. Pohst, H. Zassenhaus, Algorithmic Algebraic Number Theory, Cambridge University Press, Cambridge, 1989.

10. O. Zariski, P. Samuel, Commutative Algebra, Vol I, Van Nostrand, Princeton, New Jersey, 1958.

11. H. Zassenhaus, On Hensel Factorization II, Symposia Mathematica XV (1975), 499–513, Academic Press, London.

12. H. Zimmer, Computational Problems, Methods and Results in Algebraic Number Theory, Lecture Notes in Mathematics 262, Springer-Verlag, 1972.

Appendix I — The Major Procedures

maxord

Given p : a rational prime, f : a monic, separable polynomial in Z_p[x], m_f : the p-adic value of the discriminant of f.

Let ξ = x + fQ_p[x], so that Q_p[ξ] = Q_p[x]/fQ_p[x] = A_f . Factorize f mod p, and let g be the square-free part of f. If f satisfies the Dedekind criterion with respect to p, then the ring of coefficients in A_f of the p-radical of Z_p[ξ] is a p-maximal order. Return the basis computed by dbasis(p, f, m_f, x, f, g). If f fails to satisfy the Dedekind criterion, then Z_p[ξ] is not p-maximal. Let h be an irreducible factor of f mod p.

If h is the only irredicible factor of f mod p, then ξ is primary. Return the basis computed by nilord(p, f, m_f, h).

If h is not the only irreducible factor of f mod p, then ξ is non-primary. Return the basis computed by decomp(p, f, m_f, x, f, h).

decomp

Given p : a rational prime, f : a monic, separable polynomial in Z_p[x], m_f : the p-adic value of the discriminant of f, θ : a non-primary element of O_f , χ_θ : the characteristic polynomial of θ, ν_θ : an irreducible factor of χ_θ mod p.

Let p^{dr} = respm(f, f^′, p^{mf} ), m_r = 2d_r + 1.

Then p^{dr} O_f ⊆ Z_p[ξ], where ξ = x + fQ_p[x], and p^{dr} Z_p = (fZ_p[x] + f^′Z_p[x]) ∩ Z_p. Furthermore, if f^∗ ≡ f (mod p^{mr} Z_p[x]), then A_f � ∼= A_f .

Modulo pZ_p[x], set b_2 ≡ ν^k_θ , with k such that χ_θ ≡ b_1b_2 and gcd(b_1, b_2) ≡ 1. Compute a_1, a_2 so that a_1b_2 + a_2b_1 ≡ 1 (mod pZ_p[x]), and set e(ξ) = a_1(θ(ξ))b_2(θ(ξ)) k = 1. In the loop that follows we will always have e(ξ)(1 − e(ξ)) ∈ p^kO_f . Repeat the sequence e(ξ) ← 3e(ξ)^2 − 2e(ξ)^3 (mod p^{2k}Z_p[ξ]) k ← 2k until k ≥ m_r + d_r, then set f_1 ≡ gcdpm(f, p^{dr} (1 − e), p^{mr +dr} ), f_2 ≡ quo(f, f_1, x) modulo p^{mr} Z_p[x].

Define e_1 = e, e_2 = 1 − e_1, and for j = 1, 2 let ξ_j = x + f_j Q_p[x]. Compute the integral bases [ω_{j,1}(ξ_j ), . . .ω_{j,nj} (ξ_j )] = maxord(p, f_j, m_{fj} ).

Return the basis

[ e_1(ξ)ω_{1,1}(ξ), . . . , e_1(ξ)ω_{1,n1} (ξ), e_2(ξ)w_{2,1}(ξ), . . ., e_2(ξ)ω_{2,n2} (ξ) ].

nilord

Given p : a rational prime, f : a monic, separable polynomial in Z_p[x], m_f : the p-adic value of the discriminant of f, g : the unique irreducible factor of f mod p.

Initially set α = ξ = x + fQ_p[x].

Perform the following steps, starting over whenever α is replaced.

Compute χ_α, the characteristic polynomial of α.

Factorize χ_α mod p, and let ν_α be an irreducible factor of χ_α mod p. If χ_α has more than one irreducible factor mod p, then return the basis given by decomp(p, f, m_f, α, χ_α, ν_α).

Compute Δ_α, the discriminant of χ_α. If Δ_α = 0, replace α ← α + pξ.

If χ_α satisfies the Dedekind criterion with respect to p, then return the basis given by dbasis(p, f, m_f, α, χ_α, ν_α).

If v^∗(α) > 0, replace α ← α + 1.

Let v^∗(ν_α(α)) = L_α/M_α, with gcd(L_α, M_α) = 1, M_α > 0. Compute η_α such that v^∗(η_α(α)) = 1/M_α. If L_α > 1, replace α ← α + η_α(α).

Let φ be returned by bsrch(p, χ_α, k_α, η_α, M_α), where p^{kα} Z_p = Δ_αZ_p. If φ is non-primary, return the basis given by decomp(p, f, m_f, φ, χ_φ, ν_φ). Otherwise, replace α ← φ.

bsrch

Given p : a rational prime, f_α : the characteristic polynomial of α, k_α : the p-adic value of the discriminant of f_α, ν_α : the unique irreducible factor of f_α mod p, M_α : rational integer such that v^∗(ν_α(α)) = 1/M_α. Let n = deg(f_α) r = ⌈(d_α(n + p − 1) − 2)/(D_α(p − 1))⌉ π = ν_α(α). Initially set β ≡ π^{Mα} /p (mod p^{dα} Z_p[α]).

Perform the following sequence of tests for each choice of β.

If β is not primary, return β.

If D_β - D_α, find φ ∈ β + Z_p[α] with φ not primary or D_φ = lcm(D_α, D_β), and return φ.

If M_β - M_α, compute φ such that D_φ = D_α and M_φ = lcm(M_α, M_β), and return φ.

Set k = M_αv^∗(β), so v^∗(π^k) = v^∗(β), and set γ ≡ β/π^k (mod p^{dα} Z_p[α]).

If γ is not primary, return γ.

If D_γ - D_α, find φ ∈ γ + Z_p[α] with φ not primary or D_φ = lcm(D_α, D_γ), and return φ.

If M_γ - M_α, compute φ such that D_φ = D_α and M_φ = lcm(M_α, M_γ ), and return φ.

Set q = p^{rDα} , so that θ^q ∈ Z_p[α] whenever θ ∈ Z_p[α] + J_f , and set δ ≡ γ^q (mod p^{dα} Z_p[α]).

If δ is not primary, return δ.

If D_δ - D_α, find φ ∈ δ + Z_p[α] with φ not primary or D_φ = lcm(D_α, D_δ), and return φ.

If M_δ - M_α, compute φ such that D_φ = D_α and M_φ = lcm(M_α, M_δ), and return φ.

If δ /∈ Z_p[α], examine elements φ = γ + h(α), with h ∈ Z_p[x] reduced modulo pZ_p[x], until finding non-primary φ; then return φ.

If δ ∈ Z_p[α], set β ≡ π^k(γ − δ) (mod p^{dα} Z_p[α]) and repeat the sequence of tests.

dbasis

Given p : a rational prime, f : a monic, separable polynomial in Z_p[x], m_f : the p-adic value of the discriminant of f, α : an element of O_f , f_α : the (separable) characteristic polynomial of α, g_α : the square-free part of f_α mod p.

Let h_α = Quo(f_α, g_α, x) mod p k_α =^1_p (f_α − g_αh_α) g_β = Gcd(g_α, h_α, k_α) mod p h_β = Quo(f_α, g_β, x) mod p and define β = h_β (α) p . The coefficient ring in A_f of the p-radical of Z_p[α] is

D_α = Z_p[α] + βZ_p[α], which is an order of A_f satisfying Z_p[α] ⊆ D_α ⊆ O_f , with Z_p[α] = D_α if and only if Z_p[α] = O_f .

Moreover, if deg(g_β) = m then [D_α : Z_p[α]] = p^m. Return the basis [ 1, α, . . ., α^{n−m−1}, β, αβ, . . . , α^{m−1}β ].

gcdpm

Given f_1 : a monic polynomial in Z_p[x],

f_2 : a polynomial in Z_p[x], p^m : a power of the rational prime p.

Let n = deg(f_1). For 1 ≤ j ≤ n, set n∑ k=1 B_{j,k}x^{n−k} ≡ rem(x^{n−j} f_2, f_1, x) (mod p^mZ_p[x]).

Let I be the n × n identity matrix, and set A = ( B p^mI ) ∈ Z^{2n×n}. Compute the row-reduced Hermite normal form of A.

Determine the maximum k, 1 ≤ k ≤ n, satisfying A_{kk} ≢ 0 (mod p^m).

Return n∑ j=k A_{kj} A_{kk} x^{n−j} .

respm

Given f_1 : a monic polynomial in Z_p[x],

f_2 : a polynomial in Z_p[x], p^m : a power of the rational prime p, with p^{m+1} - resultant(f_1, f_2, x).

Compute a_1, a_2 ∈ Q_p[x] such that

a_1f_1 + a_2f_2 = 1, deg(a_1) < deg(f_2), deg(a_2) < deg(f_1). Set p^c = gcd(p^m, content(p^ma_1, x), content(p^ma_2, x)). We have p^{m−c}Z_p = (f_1Z_p[x] + f_2Z_p[x]) ∩ Z_p.

Return p^{m−c}.

Implementing the Round Four maximal order algorithm 57

Appendix II — Maple Source Listings

###
### Usage:
###
###
### > f := x^7 - 7*x + 3;
### 7
### f := x - 7 x + 3
###
### > ifactor(discrim(f,x));
### 8 8
### (3) (7)
###
### > print(maxord(3,f,8));
### 2 3 4 5
### [ 1, x, x , x , x , x ,
###
### 2 3 4 5 6
### - 1/3 x + 1/3 x - 1/3 x + 1/3 x - 1/3 x + 1/3 x ]
###
### > print(maxord(7,f,8));
### 2 3 4 5 6
### [ 1, x, x , x , x , x , x ]
###

        with(linalg,coldim,delrows,matrix,rowdim,stack,vectdim,vector):

### p-maximal order of Af

        maxord := proc (p, f, mf)
        local w, r, g, h, j;

        w := Factors(f) mod p;

        r := rowdim(matrix(w[2]));

        g := product(’w[2][j][1]’,’j’=1..r); ### g = square-free part ###
        h := w[2][1][1]; ### h = irreducible factor ###

        if dedek(p,f,g,mf) then ### p-maximal by Dedekind ###
            RETURN(dbasis(p,f,mf,x,f,g))
        elif r = 1 then ### xi is primary ###
            RETURN(nilord(p,f,mf,h))
        else ### xi is non-primary ###
            RETURN(decomp(p,f,mf,x,f,h))
        fi

        end:

58 D. FORD and P. LETARD

### Af = Af1 + Af2

        decomp := proc (p, f, mf, theta, chi, nu)
        local pdr, pmr, ph, pk, j, n1, n2, t, v1, v2,
              a1, a2, b1, b2, b3, f1, f2, e, ib1, ib2, ibas;

        pdr := respm(f,diff(f,x),p^mf); ### reduced resultant ###
        pmr := pdr*pdr*p;

        b1 := chi mod p; a2 := 0;
        b2 := 1 mod p; a1 := 1;
        b3 := nu mod p;
        while degree(b3,x) > 0 do
            b1 := Quo(b1,b3,x) mod p;
            b2 := b2*b3 mod p;
            b3 := Gcdex(b2,b1,x,’a1’,’a2’) mod p ### monic ###
        od;

        e := eleval(f,a1*b2,theta);
        e := (pdr*e mod pdr*p) / pdr;

        pk := p;
        ph := pdr*pmr;

### E(t) - e(t) belongs to p^k*Op, which is contained in p^(k-dr)*Zp[xi]

        while pk < ph do
            e := rem (e*e*(3 - 2*e), f, x);
            pk := pk*pk;
            e := (pdr*e mod pdr*pk) / pdr
        od;

        f1 := gcdpm(f,pdr*(1-e),pdr*pmr); f1 := mods (f1, pmr);
        f2 := quo(f,f1,x); f2 := mods (f2, pmr);

        n1 := degree(f1,x); n2 := degree(f2,x);
        v1 := ordp(discrim(f1,x),p); v2 := ordp(discrim(f2,x),p);
        ib1 := maxord(p,f1,v1); ib2 := maxord(p,f2,v2);

        ibas := vector(n1+n2);

        for j from 1 to n1 do
            ibas[j] := rem(pdr*e*ib1[j],f,x) mod pdr
        od;
        for j from n1+1 to n1+n2 do
            ibas[j] := rem(pdr*(1-e)*ib2[j-n1],f,x) mod pdr
        od;

        RETURN(nbasis(ibas,pdr))

        end:

Implementing the Round Four maximal order algorithm 59

### polynomial gcd mod p^m (assumes f1 monic)

        gcdpm := proc (f1, f2, pm)
        local n, a, b, h, j, k, n;

        n := degree(f1,x);
        a := matrix(n,n);

        for j from 1 to n do ### Sylvester’s matrix, mod p^m ###
            if j = 1 then
                h := rem (f2, f1, x) mod pm
            else
                h := rem (x*h, f1, x) mod pm
            fi;
            for k from 1 to n do
                a[n-j+1,k] := coeff(h,x,n-k)
            od
        od;

        a := hnfpm(a,pm);

        k := 0;
        for j from 1 to n do
        if a[j,j] mod pm <> 0 then
            k := j
        fi;
        od;

        b := sum(’(a[k,j]/a[k,k])*x^(n-j)’,’j’=k..n);

        RETURN(b)

        end:

### reduced resultant mod p^m

        respm := proc (f1, f2, pm)
        local g, a1, a2, pc;

        g := gcdex (f1, f2, x, ’a1’, ’a2’);

        a1 := pm*a1 mod pm;
        a2 := pm*a2 mod pm;

        pc := igcd (pm, content(a1,x));
        pc := igcd (pc, content(a2,x));

        RETURN(pm/pc)

        end:

60 D. FORD and P. LETARD

### Hermite normal form (n x n) mod p^m

        hnfpm := proc (a, pm)
        local b, c, g, h, j, k, m, n, q, r,
              c00, c11, c12, c13, c21, c22, c23;

        m := rowdim(a); n := coldim(a);
        b := stack(a,pm*array(1..n,1..n,identity));

        for k from 1 to m do
        for j from 1 to n do
            b[k,j] := mods (b[k,j], pm) ### Q int Zp not subs Z ###
        od
        od;

        for c from 1 to n do
            h := c + m;
            g := b[c+m,c];
            for k from c to c+m-1 do
            if b[k,c] <> 0 then
                c00 := igcd(b[k,c],b[c+m,c]);
                if c00 < g then
                    h := k; g := c00
                fi
            fi
            od;
            k := c + m;
            if h <> k then
                c00 := igcdex(b[h,c],b[k,c],’c11’,’c12’);
                c21 := -b[k,c]/c00; c22 := b[h,c]/c00;
                b[h,c] := c00;
                b[k,c] := 0;
                for j from c+1 to n do
                    c13 := c11*b[h,j] + c12*b[k,j];
                    c23 := c21*b[h,j] + c22*b[k,j];
                    b[h,j] := mods (c13, pm);
                    b[k,j] := mods (c23, pm)
                od
            fi;
            if h <> c then
            for j from c to n do
                k := b[c,j]; b[c,j] := b[h,j]; b[h,j] := k
            od
            fi;
            for k from 1 to c+m do
            if k <> c then
                r := mods (b[k,c], b[c,c]);
                q := iquo (b[k,c] - r, b[c,c]);
                if q <> 0 then
                for j from c to n do
                    b[k,j] := mods (b[k,j] - q*b[c,j], pm)
                od
                fi
            fi
            od
        od;

        RETURN(delrows(b,n+1..n+m))

        end:

Implementing the Round Four maximal order algorithm 61

### xi is primary

        nilord := proc (p, fx, mf, gx)
        local alpha, chi, nu, eta, La, Ma, w, Dchi, phi, pm;

        alpha := x; chi := fx; nu := gx; Dchi := p^mf; pm := p^(mf+1);

        while true do

            if Dchi = 0 then
                alpha := alpha + p*x
            elif dedek(p,chi,nu,ordp(Dchi,p)) then
                RETURN(dbasis(p,fx,mf,alpha,chi,nu)) ### Dedekind ###
            else
                if vstar(chi,p) > 0 then
                    alpha := alpha + 1;
                    chi := subs(x=x-1,chi);
                    nu := subs(x=x-1,nu) mod p
                fi;
                w := setup(chi,p,x,nu);
                eta := w[2]; La := w[3]; Ma := w[4];
                if La > 1 then
                    alpha := alpha + eleval(fx,eta,alpha)
                else
                    w := bsrch (p, chi, ordp(Dchi,p), eta, Ma);
                    phi := eleval(fx,w[2],alpha);
                    if w[1] = 1 then
                        RETURN(decomp(p,fx,mf,phi,w[3],w[4]))
                    else
                        alpha := phi
                    fi
                fi
            fi;

            w := factcp(fx,p,alpha); chi := w[1]; nu := w[2];
            if w[4] > 1 then
                RETURN(decomp(p,fx,mf,alpha,chi,nu))
            fi;
            Dchi := discrim(mods(chi,pm),x) mod pm;
            if Dchi = 0 then
                Dchi := discrim(chi,x)
            fi

        od

        end:

62 D. FORD and P. LETARD

### Returns
### [1, phi, chi, nu] if phi non-primary
### [2, phi, chi, nu] if D_phi > D_alpha or M_phi > M_alpha

        bsrch := proc (p, fa, ka, eta, Ma)
        local n, c, pc, pcc, beta, gamma, delta,
              j, pik, Da, Vb, w, r, field;

        Da := degree(eta,x);
        n := degree(fa,x);
        pc := respm(fa,diff(fa,x),p^ka);
        c := ordp(pc,p);
        pcc := pc*pc;

        r := 1 + trunc ( (c*(n + p - 1) - 2)/(Da*(p - 1)) );

        beta := eltpow (fa, eta, Ma) / p;

        while true do

            beta := (pc*beta mod pcc) / pc;

            w := testd (p, fa, c, Da, eta, Ma, beta);
            if w[1] < 3 then RETURN(w) fi;

            Vb := vstar(w[3],p); ### w[3] = chi ###

            pik := eltpow (fa, eta, Ma*Vb);

            gamma := rem (beta*eltinv(fa,pik), fa, x);
            gamma := (pc*gamma mod pcc) / pc;

            w := testd (p, fa, c, Da, eta, Ma, gamma);
            if w[1] < 3 then RETURN(w) fi;

            delta := eltppm (fa, p^c, gamma, p^(r*Da));
            delta := (pc*delta mod pcc) / pc;

            w := testd (p, fa, c, Da, eta, Ma, delta);
            if w[1] < 3 then RETURN(w) fi;

            field := true;
            for j from 1 to n do
            if ordp(coeff(delta,x,n-j),p) < 0 then
                field := false
            fi
            od;

            if field then
                beta := beta - rem (pik*delta, fa, x)
            else
                RETURN(csrch(p,fa,gamma))
            fi

        od

        end:

Implementing the Round Four maximal order algorithm 63

### Returns [1, theta, chi, nu] with theta non-primary

        csrch := proc (p, fa, gamma)
        local t, v, h, j, r, theta, w, b;

        b := vector(4);

        t := 0;
        while true do
            t := t + 1;
            v := t;
            h := 0;
            j := 0;
            while v <> 0 do
                r := irem(v,p);
                v := iquo(v,p);
                h := h + r*x^j;
                j := j + 1
            od;
            theta := gamma + rem (h, fa, x);
            w := factcp(fa,p,theta);
            if w[4] > 1 then
                b[1] := 1; ### non-primary ###
                b[2] := theta;
                b[3] := w[1];
                b[4] := w[2];
                RETURN(b)
            fi
        od

        end:

64 D. FORD and P. LETARD

### Returns [1, phi, chi, nu] if phi non-primary
### [2, phi, chi, nu] if D_phi = lcm(D_alpha,D_theta)

        testb := proc (p, fa, Da, theta, Dt)
        local Dat, t, v, h, j, r, phi, w, b;

        Dat := ilcm(Da,Dt);
        b := vector(4);
        t := 0;
        while true do
            t := t + 1;
            v := t;
            h := 0;
            j := 0;
            while v <> 0 do
                r := irem(v,p);
                v := iquo(v,p);
                h := h + r*x^j;
                j := j + 1
            od;
            phi := theta + rem (h, fa, x);
            w := factcp (fa, p, phi);
            if w[4] > 1 then ### phi non-primary ###
                b[1] := 1;
                b[2] := phi;
                b[3] := w[1];
                b[4] := w[2];
                RETURN(b)
            fi;
            if w[3] = Dat then ### D_phi = lcm(D_alpha,D_theta) ###
                b[1] := 2;
                b[2] := phi;
                b[3] := w[1];
                b[4] := w[2];
                RETURN(b)
            fi
        od

        end:

Implementing the Round Four maximal order algorithm 65

### Returns [1, phi, chi, nu] if phi non-primary
### [2, phi, chi, nu] if M_phi > M_alpha

        testc := proc (p, fa, c, alph2, Ma, thet2, Mt)
        local g, r, s, t, c1, c2, c3, pc, ppc, psi, phi, w, b;

        b := vector(4); pc := p^c; ppc := p*pc;

        g := igcdex (Ma, Mt, ’r’, ’s’);
        t := 0; ### r Ma + s Mt - t Ma Mt = g ###
        while r < 0 do
            r := r + Mt;
            t := t + 1
        od;
        while s < 0 do
            s := s + Ma;
            t := t + 1
        od;
        c1 := eltpow(fa,alph2,s);
        c2 := eltpow(fa,thet2,r);
        c3 := rem (c1*c2, fa, x) / p^t;
        psi := (pc*c3 mod ppc) / pc; ### psi = c3 mod p ###
        phi := x + psi;

        w := factcp (fa, p, phi);
        if w[4] > 1 then ### phi non-primary ###
            b[1] := 1;
            b[2] := phi;
            b[3] := w[1];
            b[4] := w[2];
            RETURN(b)
        else ### M_phi = lcm(M_alpha,M_theta) ###
            b[1] := 2;
            b[2] := phi;
            b[3] := w[1];
            b[4] := w[2];
            RETURN(b)
        fi

        end:

66 D. FORD and P. LETARD

### Returns [1, phi, chi, nu] if phi non-primary
### [2, phi, chi, nu] if D_phi > D_alpha or M_phi > M_alpha
### [3, phi, chi, nu] otherwise

        testd := proc (p, fa, c, Da, alph2, Ma, theta)
        local chit, nut, Dt, thet2, Mt, w, b;

        b := vector(4);

        w := factcp (fa, p, theta);
        chit := w[1]; nut := w[2]; Dt := w[3];

        if w[4] > 1 then ### theta non-primary ###
            b[1] := 1;
            b[2] := theta;
            b[3] := chit;
            b[4] := nut;
            RETURN(b)
        fi;

        if Da < ilcm(Da,Dt) then ### D_phi > D_alpha ###
            RETURN(testb(p,fa,Da,theta,Dt))
        fi;

        w := setup (fa, p, theta, nut); thet2 := w[2]; Mt := w[4];

        if Ma < ilcm(Ma,Mt) then ### M_phi > M_alpha ###
            RETURN(testc(p,fa,c,alph2,Ma,thet2,Mt))
        else
            b[1] := 3;
            b[2] := theta;
            b[3] := chit;
            b[4] := nut;
            RETURN(b)
        fi

        end:

Implementing the Round Four maximal order algorithm 67

### Factorize char poly mod p

        factcp := proc (f, p, theta)
        local w, b;

        b := vector(4);
        b[1] := chpol(f,theta); ### chi_theta ###
        w := Factors(b[1]) mod p;
        b[2] := w[2][1][1]; ### nu_theta ###
        b[3] := degree(b[2],x); ### D_theta ###
        b[4] := rowdim(matrix(w[2])); ### Nr of mod p factors ###

        RETURN(b)

        end:

### Returns [theta_1, theta_2, L_theta, M_theta]
### [1] [2] [3] [4]

        setup := proc (f, p, theta, nut)
        local b, t1, t2, v1, Lt, Mt, c, r, s;

        b := vector(4);

        t1 := eleval(f,nut,theta);
        v1 := vstar(chpol(f,t1),p);
        Lt := numer(v1);
        Mt := denom(v1);
        c := igcdex(Lt,-Mt,’r’,’s’); ### r Lt - s Mt = 1 ###
        while r <= 0 do
            r := r + Mt;
            s := s + Lt
        od;
        t2 := eltpow(f,t1,r) / p^s;

        b[1] := t1;
        b[2] := t2;
        b[3] := Lt;
        b[4] := Mt;

        RETURN(b)

        end:

68 D. FORD and P. LETARD

### evaluate g(a)

        eleval := proc (f, h, a)
        local n, g, y, k;

        g := collect(h,x);
        n := degree(g,x);
        y := 0;
        for k from n by -1 to 0 do
            y := rem (y*a + coeff(g,x,k), f, x)
        od;

        RETURN(y)

        end:

### inverse of theta in Af

        eltinv := proc (f, theta)
        local g, a, b;

        g := gcdex (theta, f, x, ’a’, ’b’);

        RETURN(a)

        end:

### Power of an element

        eltpow := proc (f, theta, k)
        local phi, psi, q, r;

        phi := 1;
        psi := theta;
        q := k;

        while q <> 0 do
            r := irem(q,2);
            if r <> 0 then
                phi := rem (phi*psi, f, x)
            fi;
            q := iquo(q,2);
            if q <> o then
                psi := rem (psi*psi, f, x)
            fi
        od;

        RETURN(phi)

        end:

Implementing the Round Four maximal order algorithm 69

### Modular power of an element

        eltppm := proc (f, pd, theta, k)
        local pdd, phi, psi, q, r;

        pdd := pd*pd;
        phi := pd;
        psi := pd*theta;
        q := k;

        while q <> 0 do
            r := irem(q,2);
            if r <> 0 then
                phi := rem (phi*psi/pd, f, x);
                phi := phi mod pdd
            fi;
            q := iquo(q,2);
            if q <> o then
                psi := rem (psi*psi/pd, f, x);
                psi := psi mod pdd
            fi
        od;

        RETURN(phi/pd)

        end:

### Dedekind test for p-maximality

        dedek := proc (p, fa, ga, mfa)
        local ha, gb;

        ha := Quo(fa,ga,x) mod p;
        gb := (fa - ga*ha)/p;
        gb := Gcd(gb,ga) mod p;
        gb := Gcd(gb,ha) mod p;
        if member (2*degree(gb,x), {0, mfa-1, mfa}) then
            RETURN(true)
        else
            RETURN(false)
        fi

        end:

70 D. FORD and P. LETARD

### Dedekind basis

        dbasis := proc (p, f, mf, alpha, fa, ga)
        local pd, n, m, i, j, k, h, a, b, ha, gb, hb;

        n := degree(f,x);
        pd := p^trunc(mf/2);

        ha := Quo(fa,ga,x) mod p;
        gb := (fa - ga*ha)/p;
        gb := Gcd(gb,ga) mod p;
        gb := Gcd(gb,ha) mod p;

        a := matrix(n,n);
        m := degree(gb,x);

        for i from 1 to n do ### Zp[a] + bZp[a] is maximal ###
            if i = 1 then
                ha := pd
            elif i = n-m+1 then
                hb := Quo(fa,gb,x) mod p;
                ha := pd*eleval(f,hb,alpha)/p
            else
                ha := rem(alpha*ha,f,x)
            fi;
            for j from 1 to n do
                a[i,j] := coeff(ha,x,n-j)
            od
        od;

        a := hnfpm(a,pd);

        b := vector(n);

        for j from 1 to n do
            b[j] := sum(’a[n+1-j,n-k]*x^k’,’k’=0..n-1) / pd
        od;

        RETURN(b)

        end:

Implementing the Round Four maximal order algorithm 71

### Normalized integral basis
        nbasis := proc (ibas, pd)
        local n, j, k, a, b, h;
        n := vectdim(ibas);
        a := matrix(n,n);
        for j from 1 to n do
        for k from 1 to n do
            a[j,k] := coeff(ibas[j],x,n-k)
        od
        od;
        a := hnfpm(a,pd);
        b := vector(n);
        for j from 1 to n do
            b[j] := sum(’a[n+1-j,n-k]*x^k’,’k’=0..j-1) / pd
        od;
        RETURN(b)
        end:

### characteristic polynomial of beta
        chpol := proc (f, beta)
        local g, y;
        g := resultant (y-beta, f, x);
        g := subs(y=x,g);
        g := collect(g,x);
        g := g/lcoeff(g,x);
        RETURN(g)
        end:

72 D. FORD and P. LETARD

### p-adic valuation

        ordp := proc (w, p)
        local nw, dw, vn, vd;

        nw := numer(w);
        dw := denom(w);
        vn := 0;
        vd := 0;

        if w = 0 then
            RETURN(0)
        else
            while nw mod p = 0 do
                nw := nw / p;
                vn := vn + 1
            od;
            while dw mod p = 0 do
                dw := dw / p;
                vd := vd + 1
            od;
            RETURN(vn-vd)
        fi

        end:

### minimum extension valuation

        vstar := proc (h, p)
        local j, m, v, w, first;

        m := degree(h,x);
        first := true;

        for j from 1 to m do
        if coeff(h,x,m-j) <> 0 then
            w := ordp(coeff(h,x,m-j),p) / j;
            if first then
                v := w
            elif w < v then
                v := w
            fi;
            first := false
        fi
        od;

        if first then
            v := 0
        fi;

        RETURN(v)

        end:

Appendix III — Examples

The different steps of the algorithm are illustrated in two first examples; the third is an extreme case for which the Round Two algorithm is faster. We also give tables of comparative performance statistics for several other examples.

All computations for these examples were done with version 1.38.40 of PARI on a dual-processor Sparc 10 system at the Centre de Recherche en Mathématiques de Bordeaux. Example 1.

Let f = x^4 + 3x^3 − x^2 + 8x + 8, a monic separable polynomial in Z[x], and let A_f = Q[x]/fQ[x]. Compute

discrim(f) = −3 · 13 · 19 · 31 · 2^5.

1.1 A 2-maximal order of A_f .

Define A_{f,2} = Q_2[x]/fQ_2[x] and let O_{f,2} be the 2-maximal order of A_{f,2}. The element ξ = x + fQ_2[x] is a non-primary element of O_{f,2}. So we construct orthogonal idempotents and decompose the algebra as the direct sum of subalgebras of lower degree.

We seek orthogonal idempotents e and 1 − e and polynomials f_1 and f_2 such that

A_{f,2} = A_{f1,2} ⊕ A_{f2,2} = Q_2[x]/f_1Q_2[x] ⊕ Q_2[x]/f_2Q_2[x]. The decomp procedure gives e = 47077x^3 + 12438x^2 + 58712x + 27857, f_1 = x^2 + 32x + 184, f_2 = x^2 + 483x + 231.

We recursively compute 2-maximal orders O_{f1,2} of A_{f1,2}, O_{f2,2} of A_{f2,2}.

1.1.1 Integral basis for O_{f1,2}.

Given f_1 = x^2 + 32x + 184 and ξ_1 = x + f_1Q_2[x], the nilord procedure constructs an Eisenstein element^3_2 (ξ_1 + 2) with minimal polynomial x^2 + 42x + 279.

So a basis for the 2-maximal order of A_{f1,2} is: [ 1,^1_2 x ] Let w_{1,f1} (ξ_1), w_{2,f1} (ξ_1) be the two vectors for this basis. 1.1.2 Integral basis for O_{f2,2}.

For f_2 = x^2 + 483x + 231 the Dedekind test gives a basis for O_{f2,2}: [1, x] . Define ξ_2 = x + f_2Q_2[x] and let w_{1,f2} (ξ_2), w_{2,f2} (ξ_2) be the two vectors for this basis.

1.1.3 Integral basis for O_{f,2}.

The 2-maximal order of A_{f,2} can be constructed directly by mapping integral bases of A_{f1,2} and A_{f2,2} into A_{f,2}. The Hermite Form of [ e(ξ)w_{1,f1} (ξ), e(ξ)w_{2,f1} (ξ), (1 − e)(ξ)w_{1,f2} (ξ), (1 − e)(ξ)w_{2,f2} (ξ) ] gives a basis for O_{f,2}: [ 1, x, x^2,^1_2 (x^3 + x^2 + x) ] . 1.2 Integral basis for A_f .

Only p = 2 divides F , so the integral basis is the 2-maximal order of A_f : [ 1, x, x^2,^1_2 (x^3 + x^2 + x) ] and the discriminant of A_f is: −3 · 13 · 19 · 31 · 2^3 = −183768.

Example 2.

Let f = x^5 + 4x^4 + 19x^3 + 3x^2 + 12x + 9, a monic separable polynomial in

Z[x], and let Af = Q[x]/fQ[x]. Compute

discrim(f) = 17 · 42239 · 2^8 · 3^4.

2.1 A 2-maximal order of A_f .

Define A_{f,2} = Q_2[x]/fQ_2[x] and let O_{f,2} be the 2-maximal order of A_{f,2}. The decomp procedure gives:

e = 52756x^4 + 32469x^3 + 21251x^2 + 56955x + 31530, f_1 = x^3 + 191x^2 + 207x + 833, f_2 = x^2 + 1861x + 713.

We recursively compute 2-maximal orders O_{f1,2} of A_{f1,2}, O_{f2,2} of A_{f2,2}. 2.1.1 A 2-maximal order of O_{f1,2}.

For f_1 = x^3 + 191x^2 + 207x + 833 the decomp procedure gives: e =^1_4 (124823x^2 + 69006x + 47623), f_{1,1} = x^2 + 162x + 1653, f_{1,2} = x + 29.

2.1.1.1 Integral basis for O_{f1,1,2}.

Given f_{1,1} = x^2+162x+1653 and ξ_{1,1} = x+f_{1,1}Q_2[x], the nilord procedure constructs an Eisenstein element^3_2 (ξ_{1,1} + 1) with minimal polynomial x^2 + 240x + 3357.

So a basis for the 2-maximal order of A_{f1,1,2} is: [ 1,^1_2 (x + 1) ] .

2.1.1.2 Integral basis for O_{f1,2,2}.

Given f_{1,2} = x + 29 the Dedekind test gives a basis for O_{f1,2,2}: [1, x] . 2.1.1.3 Integral basis for O_{f1,2}.

The Hermite Form of [ e_{1,1}(ξ_1)w_{1,f1,1} (ξ_1), e_{1,1}(ξ_1)w_{2,f1,1} (ξ_1), e_{1,2}(ξ_1)w_{1,f1,2} (ξ_1), e_{1,2}(ξ_1)w_{2,f1,2} (ξ_1) ] gives a basis for O_{f1,2}: [ 1,^1_2 (x + 1),^1_4 (x^2 − 1) ] . Let w_{1,f1} (ξ_1), w_{2,f1} (ξ_1), w_{3,f1} (ξ_1) be these three vectors.

2.1.2 Integral basis for O_{f2,2}.

For f_2 = x^2 + 1861x + 713 the Dedekind test gives a basis for O_{f1,2,2}: [1, x] . Define ξ_2 = x + f_2Q_2[x] and let w_{1,f2} (ξ_2), w_{2,f2} (ξ_2) be the two vectors for this basis.

2.1.3 Integral basis for O_{f,2}.

The 2-maximal order of A_{f,2} can be constructed directly by mapping integral bases of A_{f1,2} and A_{f2,2} into A_{f,2}.

The Hermite Form of [ e_1(ξ)w_{1,f1}(ξ), e_1(ξ)w_{2,f1} (ξ), e_1(ξ)w_{3,f1} (ξ), e_2(ξ)w_{1,f2} (ξ), e_2(ξ)w_{2,f2}(ξ) ] . gives a basis for the 2-maximal order: [ 1, x, x^2,^1_2 (x^3 + 1),^1_4 (x^4 + x^3 − x − 1) ] .

2.2 A 3-maximal order of A_f .

Define A_{f,3} = Q_3[x]/fQ_3[x] and let O_{f,3} be the 3-maximal order of A_{f,3}.

The decomp procedure gives:

e = 4755x^4 + 1610x^3 + 5814x^2 + 5955x + 2071, f_1 = x^3 + 12x + 9, f_2 = x^2 + 4x + 7.

We recursively compute 3-maximal orders O_{f1,3} of A_{f1,3}, O_{f2,3} of A_{f2,3}.

A basis for O_{f1,3} is [1, x,^1_3 x^2] and for O_{f2,3} is [1, x].

So, a basis for O_{f,3} is [ 1, x, x^2, x^3,^1_3 (x^4 + x^3 + x^2) ] . 2.3 Integral basis for A_f :

Computing HNF ( 12O_{f,2} | | 12O_{f,3})

= HNF

12 0 0 6 −3 12 0 0 0 0 0 12 0 0 −3 0 12 0 0 0 0 0 12 0 0 0 0 12 0 4 0 0 0 6 3 0 0 0 12 4 0 0 0 0 3 0 0 0 0 4

gives an integral basis for A_f : [ 1, x, x^2,^1_2 (x^3 + 1),^1_{12} (x^4 + x^3 + 4x^2 + 3x + 3) ] .

Example 3.

This example is from [Ash, Pinch & Taylor, 1991]. Let

f(x) = x^{12} − 181170x^{11} + 13676070375x^{10} − 564635734535475x^9 + 14120575648656756795x^8 − 224213861531349946866060x^7 + 2299324928127100837257833640x^6 − 15120132032108410885407953780505x^5 + 61607021939453175254804920116967515x^4 − 144536083330213614666317706146365094565x^3 + 170426077617455313511361437803852538934904x^2 − 83139235455474245627641509862888062014092560x + 12253655221465755667504199645608996691723374656.

Then f is irreducible in Z[x]. Factoring the discriminant of f requires 46.15 CPU-seconds and gives

2^{54} · 3^{210} · 61^{98} · 233^2 · 419^8 · 1627^6 · 246319^2 · 1986499^8 · 156994183^2 · 102830099^2 · 369279563^2 · 712707529^6 · 63568512603919^2.

Computing the discriminant of Q[x]/fQ[x] requires 208.41 CPU-seconds, giving 139754631175017849 = 3^6 · 61^8.

So the index is

2^{27} · 3^{102} · 61^{45} · 233 · 419^4 · 1627^3 · 246319 · 1986499^4 · 156994183 · 102830099 · 369279563 · 712707529^3 · 63568512603919.

Remark: For this example the Round Two algorithm is faster than Round Four (Round Two takes 120 CPU-seconds). However, the computation of each p-maximal order is faster with Round Four, except when p = 61. This case takes a considerable amount of time. Apparently working modulo such

Implementing the Round Four maximal order algorithm 79

a large power of 61 is slower than handling matrices of size 12^2 × 12. If a means were known to identify such cases in advance, a better algorithm (combining Round Two and Round Four) could be found.

Statistical Comparison of Round Four to Round Two.

Using the 650 test polynomials of degrees 3 through 15 from [Ford, 1978] we identified 468 pairs (f, p) for which the Dedekind test for f does not immediately give a p-maximal order. A least-squares straight line fit to the points (ln n, ln t_2/t_4) for these pairs gives the relation t_2 t_4

≈ 0.0426 n^{2.16}.

3 ∘ ∘ ∘∘∘∘ ∘∘ ∘ ∘∘ ∘

4 ∘∘∘∘∘∘∘ ∘∘∘∘ ∘∘∘∘∘

5 ∘∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘ 6 ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘ 7 ∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘∘ 8 ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘ 9 ∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘∘∘∘∘∘ 10 ∘∘∘∘∘∘∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ 11 ∘ ∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘∘ 12 ∘∘∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘ 13 ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘∘∘ ∘∘ 14 ∘ ∘∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘∘∘∘∘∘∘ ∘ ∘ 15 ∘∘ ∘∘∘∘∘∘ ∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘∘ ∘ | 0.25 | 0.5 | 1 | 2 | 4 | 8 | 16 | 32 Execution Times: n vs t_2/t_4

Comparisons of Individual Examples.

We compare the CPU times used by the Round Four and Round Two algorithms for several examples. Times are given in milliseconds, with the time required to factorize the polynomial discriminant omitted.

Polynomial Rd 4 Rd 2 Ratio

x^9 − 2x^4 − 10x^3 + x − 2 470 1550 3.3 x^9 − 2x^5 + 17x^3 + 4 270 920 3.4 x^9 − 2x^3 − 10 200 2940 14.7

x^{10} + 7x^9 − 2x^8 − 2x^7 − 3x^5 + x^4 + 1 70 810 11.6 x^{10} − 4x^9 − 8x^5 + 5x^4 + 1 70 1890 27.0 x^{10} − 2x^9 − 15 270 3900 14.4

x^{11} + x^8 − 2x^2 + 4 810 4000 4.9 x^{11} − x^6 − 2x^3 − 12x^2 − 6 630 7050 11.1 x^{11} − x^{10} − x^4 − 4 560 7150 12.7

x^{12} − 3x^9 + 4x^8 − x^6 − x^2 + 10 120 3360 28.0 x^{12} + 4x^{11} + 5x^{10} + 6x^6 − 3x^4 + 12 640 10920 17.0 x^{12} + x^9 − 9x^7 − 2x^6 − 9x^5 − 6 80 2080 26.0

x^{13} + 6x^{10} − 10x^5 + 9x^2 − 2 230 4290 18.7 x^{13} + x^{10} + x^9 − 4x^8 − x^4 + x^2 − 1 130 4730 36.4 x^{13} + x^{11} − 8 1100 17500 15.9

x^{14} − x^{12} − x^7 + 10x^5 − 4 710 14630 20.6 x^{14} + 2x^8 + 6x − 1 810 5280 6.5 x^{14} − 8x^7 + 418 550 19910 36.2

x^{15} + 4x^{11} + 12x^{10} + x^3 − 4 900 15360 17.1 x^{15} + 9x^5 + 1 1210 13030 10.8 x^{15} − 13x^5 − 2 420 17400 41.4 x^{15} − 30x^{13} + 360x^{11} − 2200x^9 + 7200x^7 70 7670 109.5 −12096x^5 + 8960x^3 − 120x − 249 x^{15} − 30x^{13} + 360x^{11} − 2200x^9 + 7200x^7 1370 11670 8.5 −12096x^5 + 8960x^3 − 120x − 257

x^{16} + 132x^{14} + 6868x^{12} + 179570x^{10} 5010 87230 17.4 +2494972x^8 + 18111820x^6 + 65000173x^4 +102234000x^2 + 46240000

x^{21} − 42x^{19} + 756x^{17} − 7616x^{15} + 47040x^{13} 670 67720 101.0 −183456x^{11} + 448448x^9 − 658944x^7 +532224x^5 − 197120x^3 + 21504x − 1691

David Ford Centre Interuniversitaire en Calcul Mathématique Algébrique Department of Computer Science Concordia University 1455 de Maisonneuve Boulevard West Montréal, Québec Canada

Pascal Letard Laboratoire d’Algorithmique Arithmétique et Expérimentale Université Bordeaux I CNRS 351 cours de la Libération 33405 Talence Cedex, France